10/24/19
Some chatting about project 2, and comparison to both section 7 of the class text and the special case of k=0.9 done in this worksheet. If you missed it, you missed it.

Now back to the phugoid without acceleration.

First, let's define the system, this time defining it in a slightly different (but equivalent) way.

```
> F := (theta, v) → [v - cos(theta)/v, -sin(theta) - R·v^2];
  Phug := [D(theta)(t) = F(theta(t), v(t))[1], D(v)(t) = F(theta(t), v(t))[2]];
```

$$F := (\theta, v) \mapsto [v + (-\cos(\theta)\, v^{-1}), -\sin(\theta) + (-R\, v^2)]$$

$$Phug := \left[D(\theta)(t) = v(t) - \frac{\cos(\theta(t))}{v(t)}, D(v)(t) = -\sin(\theta(t)) - R\, v(t)^2\right] \quad \textbf{(1)}$$

Find fixed point. Note that maple assumes that if you want to solve a pair of expressions, you want to set each to zero.
This works in our favor. Or, you could write
$solve([F(\text{theta}, v)[1] = 0, F(\text{theta}, v)[1] = 0], \{\text{theta}, v\})$
which is completely equivalent. Whatever makes you happy.

```
> solve(F(theta, v), {theta, v})
```

$$\{\theta = \arctan(-RootOf(-1 + (R^2+1)\_Z^2)\, R, RootOf(-1 + (R^2+1)\_Z^2)), v = RootOf(\_Z^2 - RootOf(-1 + (R^2+1)\_Z^2))\} \quad \textbf{(2)}$$

```
> FixptEq := convert((2), radical)
```

$$FixptEq := \left\{\theta = \arctan\left(-\sqrt{\frac{1}{R^2+1}}\, R, \sqrt{\frac{1}{R^2+1}}\right), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right\} \quad \textbf{(3)}$$

```
> FixptEq := convert((2), radical) assuming R ≥ 0
```

$$FixptEq := \left\{\theta = -\arctan(R), v = \left(\frac{1}{R^2+1}\right)^{1/4}\right\} \quad \textbf{(4)}$$

Above, we added the "
assuming $R \geq 0$" so that maple will choose the proper branch of the inverse tangent.

```
> Fixpt := eval([theta, v], FixptEq)
```

$$Fixpt := \left[-\arctan(R), \left(\frac{1}{R^2+1}\right)^{1/4}\right] \quad \textbf{(5)}$$

Now, let's examine the eigenvalues of the Jacobian at the fixed point.

```
> with(VectorCalculus) :
  with(LinearAlgebra) :
> Jack := eval(
  Jacobian(F(theta, v), [theta, v]), FixptEq)
```

$$Jack := \begin{bmatrix} -\dfrac{R}{\sqrt{R^2+1}\left(\dfrac{1}{R^2+1}\right)^{1/4}} & 1 + \dfrac{1}{\sqrt{R^2+1}\sqrt{\dfrac{1}{R^2+1}}} \\ -\dfrac{1}{\sqrt{R^2+1}} & -2R\left(\dfrac{1}{R^2+1}\right)^{1/4} \end{bmatrix} \quad \textbf{(6)}$$

Let's write a function that, given R, outputs the Jacobian at the fixedpoint for that value of R.

> $JackFun := unapply(Jack, R):$

> $JackFun(0)$

$$\begin{bmatrix} 0 & 2 \\ -1 & 0 \end{bmatrix} \quad (7)$$

> $Eigenvalues(JackFun(0))$

$$\begin{bmatrix} I\sqrt{2} \\ -I\sqrt{2} \end{bmatrix} \quad (8)$$

Here we can see that when R=0, the eigenvalues at the fixed point are purely imaginary (and complex conjugate), so this is a center.... at least in the linearization. (Compare to Exercise 24, where the system linearization at the origin is a center, but the fixed point is a spiral sink due to the behavior of the higher order terms).

Let's look at R=0.5

> $Eigenvalues(JackFun(.5))$

$$\begin{bmatrix} -0.709306206600000 + 1.31641660658469\, I \\ -0.709306206600000 - 1.31641660658469\, I \end{bmatrix} \quad (9)$$

For R=0.5, the eigenvalues are complex conjugate with negative real part, giving a spiral sink.
Let's look at a big value of R:

> $Eigenvalues(JackFun(3))$

$$\begin{bmatrix} -\frac{2\, 10^{3/4}}{5} \\ -\frac{10^{3/4}}{2} \end{bmatrix} \quad (10)$$

For R=3, both eigenvalues are real and negative, so we have a sink with two eigenvectors.

Let's try to find the general eigenvalues for general R.

> $Eigenvalues(JackFun(R))$

Error, (in evala/Sqrfree/preproc) reducible RootOf detected. Substitutions are {RootOf(_Z^4*(R^2+1)-1, index = 1) = RootOf((R^2+1)*_Z^2-RootOf(-R^2+_Z^2-1, index = 1)), RootOf(_Z^4*(R^2+1)-1, index = 1) = RootOf((R^2+1)*_Z^2+RootOf(-R^2+_Z^2-1, index = 1))}

That's awful. Basically, maple is telling us that there are a couple of branches for the eigenvalues as a function of R. This is actually not surprising, because they are complex, then collide at a specific value, and then become a pair of real numbers. But we don't really care about the general formula, we just want to know the point where they collide. Recalling the discussion last time, or in section 6.1 of the class text, we can just look at how the square of the trace compares to 4 times the determinant.

> $Trace(Jack)$

$$-\frac{R}{\sqrt{R^2+1}\left(\frac{1}{R^2+1}\right)^{1/4}} - 2R\left(\frac{1}{R^2+1}\right)^{1/4} \quad (11)$$

> $Determinant(Jack)$

$$\frac{2 R^2 \sqrt{R^2+1} \sqrt{\frac{1}{R^2+1}} + \sqrt{R^2+1} \sqrt{\frac{1}{R^2+1}} + 1}{(R^2+1) \sqrt{\frac{1}{R^2+1}}} \quad \mathbf{(12)}$$

> $solve\left(Trace(Jack)^2 = Determinant(Jack) \cdot 4\right)$

$$2\sqrt{2}, -2\sqrt{2} \quad \mathbf{(13)}$$

> $Eigenvalues(JackFun(2 \cdot \mathrm{sqrt}(2)))$

$$\begin{bmatrix} -\sqrt{2}\sqrt{3} \\ -\sqrt{2}\sqrt{3} \end{bmatrix} \quad \mathbf{(14)}$$

OK, so the transition from spiral sink to regular sink occurs at $R = 2\sqrt{2}$, where the two eigenvalues are equal to each other.
This is not obvious from just looking at the corresponding matrix.

$JackFun(2 \cdot \mathrm{sqrt}(2))$

$$\begin{bmatrix} -\frac{2\sqrt{2}\, 9^{3/4}}{9} & 2 \\ -\frac{\sqrt{9}}{9} & -\frac{4\sqrt{2}\, 9^{3/4}}{9} \end{bmatrix} \quad \mathbf{(15)}$$

So, the upshot of this discussion is that there are two ranges of R where the behavior is qualitatively similar (and two boundary cases). These are outlined in section 7 of the text.
For R=0, we have a center at the fixed point.
For $0<R<2\sqrt{2}$, the fixed point is a spiral sink
For $R=2\sqrt{2}$, the fixed point has a double eigenvalue which is negative (so it is a degenerate sink)
For $R>2\sqrt{2}$, the fixed point is a sink with two distinct eigenvalues and eigenvectors.

I drew a bunch of pictures on the board, but you can see them in the text.
-------------------------------------------------------------------------------------------
Let's look at some other aspects of these equations vis a vis glider flight.

> $with(DEtools):$

> $R := 0;$

$DEplot\Big(Phug, [\theta, v], t = -1..10,$

$[[\theta(0) = 0, v(0) = 1.2], [\theta(0) = 0, v(0) = 2],$

$[\theta(0) = 0, v(0) = 1.7], [\theta(0) = 0, v(0) = 1.75], [\theta(0) = 0, v(0) = 1.8]],$

$linecolor = [black\$2, blue, red, green],$

$\theta = -\frac{\mathrm{Pi}}{2} .. \frac{3 \cdot \mathrm{Pi}}{2}, v = 0..3,\ size = [.5, .4], stepsize = .05\Big)$

$$R := 0$$




\>

How do we find the stalling solution?

Some discussion here about trying to find the stall solution by trying values that are start with theta=0 and a velocity that is too small (the blue one), too large (the green one), then trying halfway between (the red one), then repeating.....

But here's another way: start near the stalling solution (theta=Pi/2, v almost 0) and run time backwards. (this is in red below)

\> $DEplot\Big(Phug, [\text{theta}, v], t = -1..10,$

$\Big[[\text{theta}(0) = 0, v(0) = 1.2], [\text{theta}(0) = 0, v(0) = 2],$

$[\text{theta}(0) = 0, v(0) = 1.7], \left[\text{theta}(5) = \frac{\text{Pi}}{2}, v(5) = 0.0001\right], [\text{theta}(0) = 0, v(0) = 1.8]\Big],$

$linecolor = [black\$2, blue, red, green],$

$\text{theta} = -\frac{\text{Pi}}{2}..\frac{3 \cdot \text{Pi}}{2}, v = 0..3, \ size = [.5, .4], stepsize = .05\Big)$




```
> zoom(%,-.1..0.1, 1.69..1.8)
```




```
> zoom(%,-.005..0.005, 1.730..1.732)
```




So we can see that theta=0, v=1.7317 is pretty close to a solution which stalls.

We'll discuss a better way next time.